Computing Areas of Regions With Discretely Defined Boundaries

Harry L. Stern
Polar Science Center
1013 N.E. 40th Street
Seattle, WA 98195

1. Introduction

It is well known that the area of a region in the plane can be computed by an appropriate integration around the boundary of the region [e.g. Hildebrand, page 306]. If the boundary is defined by a sequence of points connected by straight lines (a polygon), the parametric representation of the boundary is particularly simple, and an explicit formula for the area can be derived. Using Stokes' Theorem, this idea can be extended to derive area formulas for regions on non-planar surfaces whose boundaries are defined by a sequence of points connected by appropriate curves. In this note we present exact area formulas for regions in the plane and regions on the sphere whose boundaries are defined by such discrete sets of points.

An application of these formulas arises in computing the area of a region on a map. Suppose that the boundary of the region of interest is traced by an encoding device that records its coordinates, relative to some user-defined (x, y) system, in a computer file. Such a file may contain hundreds or thousands of coordinate pairs. If the map covers a relatively small region, the surface of the earth can be approximated locally by a plane, and the area computed directly from the (x, y) coordinate pairs. If the map covers a large region, the earth can be approximated by a sphere. The (x, y) coordinate pairs are then converted to latitude and longitude using the appropriate map projection equations, and the area on the sphere is computed.

The usual method for computing area is to divide up the two dimensional surface into a large number of small cells, and to add up the areas of those cells that lie inside the boundary of the region. This method is computationally slow, because every cell must be tested for inclusion in the region, and because high accuracy requires a small cell size. In contrast, the formulas derived here, besides being exact, are quickly evaluated on a computer because the computation is proportional to the number of boundary points. The two dimensional area calculation is reduced to a one dimensional boundary calculation.

The next section outlines the general mathematical formulation. Sections 3 and 4 give explicit results for the plane and sphere. A numerical example and concluding remarks are presented in the last section.

2. General Formulation

Stokes' theorem says

$$\iint_S (\nabla \times \mathbf{F}) \cdot \hat{\mathbf{n}} \, dA = \oint_C \mathbf{F} \cdot \frac{d\mathbf{R}}{dt} \, dt \tag{1}$$

where S is the region of a surface bounded by the curve C, $\hat{\mathbf{n}}$ is the unit outward normal on the surface, $\mathbf{R}(t)$ is a parametric representation of C, and $\mathbf{F}$ is an arbitrary vector field. We suppose that the surface is specified in some way (e.g. $x^2 + y^2 + z^2 = 1$ for the unit sphere), so that the unit outward normal $\hat{\mathbf{n}}$ can be determined (e.g. $\hat{\mathbf{n}} = x\hat{\imath} + y\hat{\jmath} + z\hat{\mathbf{k}}$ for the unit sphere). We then choose any vector field $\mathbf{F}$ such that the integrand on the left hand side of (1) is unity in S:

$$(\nabla \times \mathbf{F}) \cdot \hat{\mathbf{n}} = 1. \tag{2}$$

With F determined (though not uniquely) by equation (2), the left hand side of (1) simply reduces to the area of S, giving

$$A = \oint_C \mathbf{F} \cdot \frac{d\mathrm{R}}{dt}\, dt. \tag{3}$$

In order to evaluate the integrand on the right hand side of (3), we need a description of C. Suppose that N points on the surface are given, $\mathbf{P}_1$, $\mathbf{P}_2$, ..., $\mathbf{P}_N$, and that C is defined by connecting these points in sequence, returning to $\mathbf{P}_1$ (define $\mathbf{P}_{N+1} \equiv \mathbf{P}_1$). On each segment, from $\mathbf{P}_k$ to $\mathbf{P}_{k+1}$, let $\mathbf{R}_k(t)$ be a parametric representation of the connecting curve. There are many possible connecting curves to choose from, but the most natural choice is the geodesic, the curve of minimum length (e.g. a straight line in the plane, a great circle on the sphere). The geodesics can be found in principle from a description of the surface (for example, Weinstock pages 61-62). The collection of the N geodesics $\mathrm{R}_k(t)$ connecting the N points $\mathbf{P}_1$, $\mathbf{P}_2$, ..., $\mathbf{P}_N$, constitutes the parametric description $\mathrm{R}(t)$ of C on the right hand side of (3).

Now that we have specified how to construct the integral in (3) as a sum of integrals along the N connecting geodesics, the area formula can be written more explicitly as

$$A = \sum_{k=1}^{N} \int_0^{L_k} \mathbf{F}(s) \cdot \frac{d\mathrm{R}_k}{ds}\, ds \tag{4}$$

where s is the arc length parameter along the geodesic $\mathrm{R}_k(s)$, and L_k is the total arc length of the k-th segment. The geodesics need not necessarily be parameterized by arc length, but this is what we have used in the sections that follow.

The determination in principle of all quantities is now complete. To summarize the steps: Given a surface and a set of points $\mathbf{P}_k, k = 1, 2, \ldots, N$ that defines the boundary of a region on the surface,

(1) Find the unit outward normal on the surface, $\hat{\mathbf{n}}$;

(2) Find a vector field $\mathbf{F}$ that satisfies equation (2): $(\nabla \times \mathbf{F}) \cdot \hat{\mathbf{n}} = 1$;

(3) Find a parameterization $\mathrm{R}_k(s)$ of the geodesic from point $\mathbf{P}_k$ to $\mathbf{P}_{k+1}$;

(4) Form the integrand in equation (4) and do the integration;

(5) Sum the contributions in (4) to get the area of the region.

Some specific cases follow.

3. The Plane

In the plane $z = 0$, the unit outward normal is $\hat{\mathbf{n}} = (0, 0, 1)$ and the condition (2) on the components (F_1, F_2, F_3) of $\mathbf{F}$ is

$$\frac{\partial F_2}{\partial x} - \frac{\partial F_1}{\partial y} = 1. \tag{5}$$

We choose $F_1 = -y/2$ and $F_2 = x/2$. The geodesics $\mathbf{R}(s) = (x(s), y(s), 0)$ are straight lines, and the integral in equation (4) becomes

$$I_k = \int_0^{L_k} \frac{1}{2}\left(x\frac{dy}{ds} - y\frac{dx}{ds}\right) ds. \tag{6}$$

Let the boundary points $\mathbf{P}_k$ have coordinates (x_k, y_k). The parametric equations for the boundary segment connecting $\mathbf{P}_k$ and $\mathbf{P}_{k+1}$ (of length L_k) are

$$x(s) = x_k + \frac{s}{L_k}(x_{k+1} - x_k) \qquad y(s) = y_k + \frac{s}{L_k}(y_{k+1} - y_k). \tag{7}$$

Substituting these expressions into equation (6) with $\Delta x = x_{k+1} - x_k$ and $\Delta y = y_{k+1} - y_k$ gives

$$\begin{aligned} I_k &= \frac{1}{2}\int_0^{L_k}\left\{\left(x_k + \frac{s\,\Delta x}{L_k}\right)\left(\frac{\Delta y}{L_k}\right) - \left(y_k + \frac{s\,\Delta y}{L_k}\right)\left(\frac{\Delta x}{L_k}\right)\right\} ds \\ &= \frac{1}{2}\int_0^{L_k}\left\{\frac{x_k\Delta y}{L_k} - \frac{y_k\Delta x}{L_k}\right\} ds \\ &= \frac{1}{2}(x_k\Delta y - y_k\Delta x) \\ &= \frac{1}{2}(x_k y_{k+1} - y_k x_{k+1}). \end{aligned} \tag{8}$$

It follows that the area of the polygon in the plane whose vertexes are the points (x_k, y_k) is

$$A = \frac{1}{2}\sum_{k=1}^{N}(x_k y_{k+1} - y_k x_{k+1}) \tag{9}$$

where $x_{N+1} \equiv x_1$, $y_{N+1} \equiv y_1$, and the points (x_k, y_k) trace the boundary in a counterclockwise sense. If the order of the points is reversed, the negative of the area will result.

4. The Sphere

Without loss of generality we consider the unit sphere. It will be convenient to use both rectangular and spherical coordinates. The longitude θ, measured positive eastward, and latitude ϕ, measured positive northward, are related to x, y, z via

$$x = \cos\phi\,\cos\theta \quad y = \cos\phi\,\sin\theta \quad z = \sin\phi \tag{10}$$

and the unit vectors in the θ, ϕ, and radial directions are related to the rectangular unit vectors $\hat{\mathbf{i}}$, $\hat{\mathbf{j}}$, $\hat{\mathbf{k}}$ via

$$\hat{\mathbf{u}}_\theta = (-\sin\theta)\hat{\mathbf{i}} + (\cos\theta)\hat{\mathbf{j}} = \frac{-y}{\sqrt{1-z^2}}\hat{\mathbf{i}} + \frac{x}{\sqrt{1-z^2}}\hat{\mathbf{j}} \tag{11a}$$

$$\begin{aligned} \hat{\mathbf{u}}_\phi &= (\sin\phi\,\cos\theta)\hat{\mathbf{i}} + (\sin\phi\,\sin\theta)\hat{\mathbf{j}} + (-\cos\phi)\hat{\mathbf{k}} \\ &= \frac{xz}{\sqrt{1-z^2}}\hat{\mathbf{i}} + \frac{yz}{\sqrt{1-z^2}}\hat{\mathbf{j}} - \sqrt{1-z^2}\,\hat{\mathbf{k}} \end{aligned} \tag{11b}$$

$$\hat{\mathbf{u}}_r = (\cos\phi\cos\theta)\hat{\mathbf{i}} + (\cos\phi\sin\theta)\hat{\mathbf{j}} + (\sin\phi)\hat{\mathbf{k}} = x\hat{\mathbf{i}} + y\hat{\mathbf{j}} + z\hat{\mathbf{k}}. \tag{11c}$$

The unit outward normal on the sphere is just the unit radial vector $\hat{\mathbf{u}}_r$. With the vector $\mathbf{F}$ written in terms of its spherical components $\mathbf{F} = F_\theta\hat{\mathbf{u}}_\theta + F_\phi\hat{\mathbf{u}}_\phi + F_r\hat{\mathbf{u}}_r$, the condition (2) becomes [Hildebrand]

$$(\nabla\times\mathbf{F})\cdot\hat{\mathbf{u}}_r = \frac{1}{\cos\phi}\left[\frac{\partial}{\partial\theta}(F_\phi) - \frac{\partial}{\partial\phi}(\cos\phi\ F_\theta)\right] = 1. \tag{12}$$

This is most naturally satisfied if we take

$$\frac{\partial}{\partial\phi}(\cos\phi\ F_\theta) = -\cos\phi \qquad \frac{\partial}{\partial\theta}(F_\phi) = 0 \tag{13}$$

or

$$F_\theta = -\tan\phi + \frac{g(\theta)}{\cos\phi} \qquad F_\phi = h(\phi) \tag{14}$$

where g is an arbitrary function of θ, and h is an arbitrary function of ϕ. No radial dependence has been introduced into g and h because we are only interested in the values of $\mathbf{F}$ on the surface $r = constant$. Also, the radial component of $\mathbf{F}$, F_r, is of no consequence: any tangent vector to the sphere, $d\mathbf{R}/dt$, has no radial component, so the dot product $\mathbf{F}\cdot d\mathbf{R}/dt$ annihilates any radial contribution from $\mathbf{F}$. Therefore we take $F_r = 0$.

Now that $\mathbf{F}$ is determined (up to two arbitrary functions), we turn to the parameterization of the boundary. We suppose that N pairs of longitude/latitude coordinates are given, namely θ_k, ϕ_k for $k = 1, 2, \ldots, N$ (with $\theta_{N+1} \equiv \theta_1$ and $\phi_{N+1} \equiv \phi_1$), that form the boundary of the region when the points are connected in the given order. The boundary points will also be denoted by $\mathbf{P}_k$, and by their rectangular coordinates (x_k, y_k, z_k). We can use equation (10) to go from spherical to rectangular coordinates.

To simplify the notation a bit, let $k = 1$ and consider the great circular arc from $\mathbf{P}_1$ to $\mathbf{P}_2$. Let Δ represent the angle subtended at the center of the sphere by $\mathbf{P}_1$ and $\mathbf{P}_2$. Then Δ satisfies $\cos\Delta = \mathbf{P}_1\cdot\mathbf{P}_2$ since all the $\mathbf{P}_k$ are unit vectors. Note that Δ is also the length of the arc from $\mathbf{P}_1$ to $\mathbf{P}_2$. Let α be the arc length parameter along the great circle from $\mathbf{P}_1$ to $\mathbf{P}_2$, and let $\mathbf{R}(\alpha)$ be the position vector along the great circle. Since $\mathbf{R}(\alpha)$ lies in the plane spanned by $\mathbf{P}_1$ and $\mathbf{P}_2$, we can write

$$\mathbf{R}(\alpha) = A(\alpha)\mathbf{P}_1 + B(\alpha)\mathbf{P}_2 \tag{15}$$

where $A(\alpha)$ and $B(\alpha)$ are determined from the following two conditions:

(1) $\mathbf{R}(\alpha)$ lies on the unit sphere: $\mathbf{R}\cdot\mathbf{R} = 1$;

(2) The angle between $\mathbf{P}_1$ and $\mathbf{R}(\alpha)$ is α: $\mathbf{P}_1\cdot\mathbf{R} = \cos\alpha$. Using equation (15) for $\mathbf{R}$ and the fact that $\mathbf{P}_1\cdot\mathbf{P}_2 = \cos\Delta$, these conditions translate into

$$A^2 + B^2 + 2AB\cos\Delta = 1 \qquad A + B\cos\Delta = \cos\alpha \tag{16}$$

respectively. Solving for A and B, we find

$$\mathbf{R}(\alpha) = \frac{\sin(\Delta-\alpha)}{\sin\Delta}\mathbf{P}_1 + \frac{\sin(\alpha)}{\sin\Delta}\mathbf{P}_2. \tag{17}$$

This is the arc length parameterization for the great circle through $\mathbf{P}_1$ and $\mathbf{P}_2$.

With $\mathbf{R}(\alpha)$ determined, the next step is to compute $d\mathbf{R}/d\alpha$ and then $\mathbf{F}\cdot d\mathbf{R}/d\alpha$. Computation of $d\mathbf{R}/d\alpha$ is simple, but we want to express the result in terms of the unit vectors $\hat{\mathbf{u}}_\theta$ and $\hat{\mathbf{u}}_\phi$, to facilitate taking the dot product with $\mathbf{F}$. Toward this end, write

$$\frac{d\mathbf{R}}{d\alpha} = G(\alpha)\hat{\mathbf{u}}_\theta + H(\alpha)\hat{\mathbf{u}}_\phi \tag{18}$$

where $G(\alpha)$ and $H(\alpha)$ are determined as follows. Let $'$ denote $d/d\alpha$ and write $\mathbf{R}(\alpha) = (x(\alpha), y(\alpha), z(\alpha))$ where the functions x, y, z are given explicitly by the components of equation (17). Then the dot product of equation (18) with $\hat{\mathbf{u}}_\theta$ and $\hat{\mathbf{u}}_\phi$ gives, respectively, $G(\alpha)$ and $H(\alpha)$. Using equations (11a,b) to express $\hat{\mathbf{u}}_\theta$ and $\hat{\mathbf{u}}_\phi$ in terms of $\hat{\imath}, \hat{\jmath}, \hat{\mathbf{k}}$ we have

$$\begin{aligned} G(\alpha) &= \mathbf{R}'\cdot\hat{\mathbf{u}}_\theta \\ &= (x'\hat{\imath} + y'\hat{\jmath} + z'\hat{\mathbf{k}})\cdot\left[\frac{-y}{\sqrt{1-z^2}}\hat{\imath} + \frac{x}{\sqrt{1-z^2}}\hat{\jmath}\right] \\ &= \frac{xy' - yx'}{\sqrt{1-z^2}} \end{aligned} \tag{19}$$

and

$$\begin{aligned} H(\alpha) &= \mathbf{R}'\cdot\hat{\mathbf{u}}_\phi \\ &= (x'\hat{\imath} + y'\hat{\jmath} + z'\hat{\mathbf{k}})\cdot\left[\frac{xz}{\sqrt{1-z^2}}\hat{\imath} + \frac{yz}{\sqrt{1-z^2}}\hat{\jmath} - \sqrt{1-z^2}\,\hat{\mathbf{k}}\right] \\ &= \frac{x'xz}{\sqrt{1-z^2}} + \frac{y'yz}{\sqrt{1-z^2}} - z'\sqrt{1-z^2} \\ &= \frac{(x'x + y'y + z'z)z - z'}{\sqrt{1-z^2}} \\ &= \frac{-z'}{\sqrt{1-z^2}} \end{aligned} \tag{20}$$

where the last step follows because $(x'x + y'y + z'z)$ is the derivative of the constant $(x^2 + y^2 + z^2)/2$.

Using equations (14) for the components of $\mathbf{F}$ and converting from θ, ϕ to x, y, z gives

$$\mathbf{F} = \left[\frac{-z}{\sqrt{1-z^2}} + \frac{g(\theta)}{\sqrt{1-z^2}}\right]\hat{\mathbf{u}}_\theta + h(\phi)\hat{\mathbf{u}}_\phi. \tag{21}$$

Using the components of $d\mathbf{R}/d\alpha$ from equations (19) and (20), we have

$$\mathbf{F}\cdot\frac{d\mathbf{R}}{d\alpha} = \left[\frac{xy' - yx'}{\sqrt{1-z^2}}\right]\left[\frac{-z}{\sqrt{1-z^2}} + \frac{g(\theta)}{\sqrt{1-z^2}}\right] - \frac{z'h(\phi)}{\sqrt{1-z^2}} \tag{22}$$

This is the integrand for the segment of the boundary integral from $\mathbf{P}_1$ to $\mathbf{P}_2$. Integration is with respect to α, from $\alpha = 0$ to $\alpha = \Delta$. The variables x, y, z and their derivatives (with respect to α) x', y', z' are all functions of α, as given by the components of equation (17).

We can choose the functions g and h to simplify equation (22). Nothing is gained by retaining the last term, so we take $h \equiv 0$. This simplifies the integrand to

$$\mathbf{F} \cdot \frac{d\mathbf{R}}{d\alpha} = \frac{(xy' - yx')(g(\theta) - z)}{1 - z^2} \tag{23}$$

Notice the potential singularities at $z = \pm 1$, i.e. the North Pole and the South Pole. Writing the denominator as $1 - z^2 = (1 - z)(1 + z)$, we see that if $g \equiv 1$ we remove the singularity at $z = 1$, and if $g \equiv -1$ we remove the singularity at $z = -1$. We must not put $g = 0$, since then $\mathbf{F}$ would vanish everywhere on the equator, violating equation (2) there. This would lead to a value of zero for the areas of the northern and southern hemispheres. In the following development we take $g \equiv 1$. In case one of the $\mathbf{P}_k$ is the South Pole, g should be replaced by -1.

We can now write the first term in the area summation of equation (4) as

$$J_1 = \int_0^{\Delta} \frac{xy' - yx'}{1 + z} \, d\alpha. \tag{24}$$

Notice the similarity to the expression for the plane, equation (6). We have explicit expressions for x, y, z, x', y' from the components of equation (17) and its derivatives, namely

$$x = \frac{\sin(\Delta - \alpha)}{\sin \Delta} x_1 + \frac{\sin(\alpha)}{\sin \Delta} x_2 \tag{25a}$$

$$x' = \frac{-\cos(\Delta - \alpha)}{\sin \Delta} x_1 + \frac{\cos(\alpha)}{\sin \Delta} x_2 \tag{25b}$$

and similar equations for y, y' and z, z'. Substituting these expressions into equation (24) and using standard trigonometric identities leads to

$$J_1 = (x_1 y_2 - y_1 x_2) \int_0^{\Delta} \frac{d\alpha}{\sin \Delta + z_1 \sin(\Delta - \alpha) + z_2 \sin \alpha}. \tag{26}$$

Recalling that this is the contribution to the area summation from the segment $k = 1$ between $\mathbf{P}_1$ and $\mathbf{P}_2$, we can write the total area as

$$A = \sum_{k=1}^{N} (x_k y_{k+1} - y_k x_{k+1}) J_k \tag{27}$$

where the terms J_k are the integrals

$$J_k = \int_0^{\Delta_k} \frac{d\alpha}{\sin(\Delta_k) + z_k \sin(\Delta_k - \alpha) + z_{k+1} \sin \alpha}, \tag{28}$$

and Δ_k comes from $\cos(\Delta_k) = \mathbf{P}_k \cdot \mathbf{P}_{k+1}$. The integral can be put into a standard form and explicitly integrated with the substitution $w = e^{i\alpha}$. Under this transformation, $d\alpha = dw/(iw)$, $\sin\alpha = (w - w^{-1})/2i$, and the integral becomes

$$J_k = \int_1^{e^{i\Delta}} \frac{2 \, dw}{aw^2 + 2bw + c} \tag{29}$$

where

$$a = z_{k+1} - z_k e^{-i\Delta} \quad b = i \sin\Delta \quad c = z_k e^{i\Delta} - z_{k+1}. \tag{30}$$

The subscript k on Δ has been dropped to reduce notational clutter.

The value of J_k depends on the sign of the discriminant $D = b^2 - ac$, or

$$D = z_k^2 + z_{k+1}^2 - 2z_k z_{k+1}\cos\Delta - \sin^2\Delta. \tag{31}$$

The three cases are [Marsden, Appendix A]

$$J_k = \begin{cases} \frac{1}{\sqrt{D}} \ln\left[\frac{aw+b-\sqrt{D}}{aw+b+\sqrt{D}}\right] & (D > 0) \\ \frac{2}{\sqrt{-D}} \arctan\left[\frac{aw+b}{\sqrt{-D}}\right] & (D < 0) \\ \frac{-2}{aw+b} & (D = 0) \end{cases} \tag{32}$$

where the expressions must be evaluated between the upper and lower limits of $w = e^{i\Delta}$ and $w = 1$. The imaginary parts of the resulting complex expressions are zero, as they must be since the original integrand and limits are real. Algebraic simplification leads us to define

$$Q = z_k + z_{k+1} + 1 + \cos\Delta \tag{33}$$

in terms of which the expressions for J_k become

$$J_k = \begin{cases} \frac{1}{\sqrt{D}} \ln\left[\frac{Q+\sqrt{D}}{Q-\sqrt{D}}\right] & (D > 0) \\ \frac{2}{\sqrt{-D}} \arctan\left[\frac{\sqrt{-D}}{Q}\right] & (D < 0) \\ \frac{Q}{(1+z_k)(1+z_{k+1})(1+\cos\Delta)} & (D = 0) \end{cases} \tag{34}$$

This completes the determination of the terms in the area formula (27). We will now summarize the steps and put them in an algorithmic format.

Problem:

Given a sequence of (longitude,latitude) coordinates on the unit sphere, (θ_k, ϕ_k), $k = 1, 2, \ldots, N$, find the area of the region that is enclosed when the points are connected in sequence by arcs of great circles.

Solution:

(1) Set the running sum to 0 and set k to 1.

(2) Compute $\cos\Delta = \mathbf{P}_k \cdot \mathbf{P}_{k+1}$ either from $x_k x_{k+1} + y_k y_{k+1} + z_k z_{k+1}$ or from $\cos\phi_k \cos\phi_{k+1} \cos(\theta_{k+1} - \theta_k) + \sin\phi_k \sin\phi_{k+1}$. Notice that we won't ever need Δ by itself, just its cosine.

(3) Compute Q from (33): $Q = z_k + z_{k+1} + 1 + \cos\Delta$ or $Q = \sin\phi_k + \sin\phi_{k+1} + 1 + \cos\Delta$.

(4) Compute the discriminant D from (31): $D = z_k^2 + z_{k+1}^2 - 2z_k z_{k+1}\cos\Delta - \sin^2\Delta$ or $D = (\sin\phi_k + \sin\phi_{k+1})^2 - (1 + \cos\Delta)(1 - \cos\Delta + 2\sin\phi_k \sin\phi_{k+1})$.

(5) Compute the integral contribution J_k in the area formula (27), using the appropriate form of equation (34).

(6) Compute the first factor in the area formula (27), $x_k y_{k+1} - y_k x_{k+1}$ or $\cos\phi_k \cos\phi_{k+1} \sin(\theta_{k+1} - \theta_k)$.

(7) Multiply together the results of steps 5 and 6 to get the k-th term in the summation of (27), and add this to the running sum.

(8) If k is less than N then increment k and go to step 2.

A computer program that implements the above algorithm is given in the appendix.

5. Numerical Example and Remarks

It is of interest in Arctic oceanography to calculate the areas of the watersheds that drain into the Arctic Ocean. The boundary of the Asian watershed that drains into the Arctic Ocean was digitized from a Mercator map of the world by tracing its circumference with an encoding device. This produced a computer file with 672 (x, y) coordinate pairs, in which the x axis coincided with the equator, the y axis coincided with the Greenwich Meridian, and the unit of length was chosen to be one degree of longitude on the equator. These (x, y) map coordinates are related to longitude θ and latitude ϕ by [Snyder]

$$x = \frac{180}{\pi}\theta \qquad y = \frac{180}{\pi}\ln\left[\arctan\left(\frac{\phi}{2} + \frac{\pi}{4}\right)\right] \tag{35}$$

where θ and ϕ are in radians. Inverting these relations and substituting the (x, y) map coordinates gives a sequence (θ_k, ϕ_k), $k = 1$ to 672, of points on the sphere that defines the boundary of the watershed.

At first a simple integration program was written in which the region lying between the minimum and maximum latitudes and longitudes of the watershed was divided into differential elements of size $\Delta\phi$ by $\Delta\theta$. The area of the watershed was calculated as $\sum \cos\phi\, \Delta\phi\, \Delta\theta$ where the summation was taken over all elements inside the watershed boundary. With each degree of latitude and longitude divided into 32 parts, this amounted to 5,918,720 elements, of which 2,516,738 were found to lie within the watershed. The program required more than 51 hours of elapsed time on a Sun workstation to arrive at the area, $1.424 \times 10^7\ \text{km}^2$.

This dismal performance led to the derivation of the formulas in this work. Using the same 672 coordinates for input, the program in the appendix arrived at the same answer in about two seconds. The 5.9 million complicated comparisons in the first program were replaced by 672 iterations of simple calculations.

Of course in any real physical problem such as the one described here, there are sources of error such as uncertainty in the exact location of the boundary, inadequate representation of the boundary by too few points, and the non-sphericity of the earth. These problems can be dealt with by acquiring better maps, digitizing the boundary with more points, and modifying the formulas here to take into account the flattening of the earth at the poles, which introduces a correction on the order of three parts per thousand.

Acknowledgment

This work was supported by NASA Grant NAGW 2513. Thanks also to Erika Dade for bringing this problem to my attention and doing the original watershed calculations.

Appendix – Fortran Program

```
      program area
      implicit undefined (a-z)
c
c______________________________________________________________________
c
c     Read a sequence of (longitude,latitude) coordinates.
c     Compute the area on the unit sphere that is enclosed by connecting
c     these points in sequence with arcs of great circles.
c
c     Refer to ''Computing Areas of Regions with Discretely Defined
c     Boundaries".
c
c______________________________________________________________________
c
c     Constants.
c
      real pi, piOver180
      parameter (pi = 3.14159265358979, piOver180 = pi / 180.0)
c
c     Parameters.
c
      integer maxPoints
      parameter (maxPoints = 1000)
c
c     Mean radius of earth in kilometers.
c
      real Rearth
      parameter (Rearth = 6371.2)
c
c     Variables.
c
      integer n, k
      real sum, first, integral, cosDelta, D, Q, R
      real cosPhiK, cosPhiK1, sinPhiK, sinPhiK1
      real phi(maxPoints), theta(maxPoints)
      character*14 filename
c
c     Read number of lon/lat coordinate pairs, and
c     the name of the file containing those coordinates.
c
      read(5,*) n, filename
c
c     Read the coordinates.  Longitude is first.  Both in degrees.
c
```

```
      open(1, file=filename)
      read(1,*) (theta(k),phi(k), k=1,n)
      close(1)
c
c     Convert to radians.
c
      do 10 k=1,n
        phi(k) = phi(k) * piOver180
        theta(k) = theta(k) * piOver180
   10 continue
c
c     Make the sequence of coordinates cyclic.
c
      phi(n+1) = phi(1)
      theta(n+1) = theta(1)
c
c     Initialize for the summation.
c
      sum = 0.0
      cosPhiK1 = cos(phi(1))
      sinPhiK1 = sin(phi(1))
c
      do 20 k=1,n
c
c       Previous "k+1" values become new "k" values.
c
        cosPhiK = cosPhiK1
        sinPhiK = sinPhiK1
c
c       Get new "k+1" values.
c
        cosPhiK1 = cos(phi(k+1))
        sinPhiK1 = sin(phi(k+1))
c
c       Compute first factor in k-th term of summation.
c
        first = cosPhiK * cosPhiK1 * sin(theta(k+1)-theta(k))
c
c       Compute integral in k-th term of summation.
c       First get cosine of delta, then discriminant, then Q.
c
        cosDelta = cosPhiK * cosPhiK1 * cos(theta(k+1)-theta(k))
     .       + sinPhiK * sinPhiK1
        D = (sinPhiK + sinPhiK1)**2
     .       - (1.0+cosDelta)*(1.0-cosDelta  +2.0*sinPhiK*sinPhiK1)
        Q = sinPhiK + sinPhiK1 + 1.0 + cosDelta
```

```
c
      if (D .gt.  0.0) then
        R = sqrt (D)
        integral = alog ( (Q+R)/(Q-R) ) / R
      else if (D .lt.  0.0) then
        R = sqrt (-D)
        integral = 2.0 * atan ( R/Q ) / R
      else
        integral = Q / ((1.0+sinPhiK) *(1.0+sinPhiK1)*(1.0+cosDelta))
      endif
c
c      Accumulate sum and go on to next segment.
c
      sum = sum + first * integral
c
   20 continue
c
c    Write results and stop.
c
      write(6,90) sum, sum/(4.0*pi), sum*Rearth*Rearth
c
      stop
   90 format(1x, 'area (on unit sphere) = ', e14.6,
     .    /1x, 'area / (4*pi) = ', e14.6,
     .    /1x, 'area (km**2 on earth) = ', e14.6)
      end
```

References

(1) Francis B. Hildebrand, "Advanced Calculus for Applications", Prentice-Hall, 1976.

(2) J. E. Marsden and A. J. Tromba, "Vector Calculus", W. H. Freeman & Co., 1976.

(3) John P. Snyder, "Map Projections - A Working Manual", U. S. Geological Survey Professional Paper 1395, U. S. Government Printing Office, 1987.

(4) Robert Weinstock, "Calculus of Variations", Dover Publications, 1974.